

**meprobit** — Multilevel mixed-effects probit regression

|                             |                                      |                                |                                      |
|-----------------------------|--------------------------------------|--------------------------------|--------------------------------------|
| <a href="#">Description</a> | <a href="#">Quick start</a>          | <a href="#">Menu</a>           | <a href="#">Syntax</a>               |
| <a href="#">Options</a>     | <a href="#">Remarks and examples</a> | <a href="#">Stored results</a> | <a href="#">Methods and formulas</a> |
| <a href="#">References</a>  | <a href="#">Also see</a>             |                                |                                      |

## Description

`meprobit` fits mixed-effects models for binary or binomial responses. The conditional distribution of the response given the random effects is assumed to be Bernoulli, with success probability determined by the standard normal cumulative distribution function.

## Quick start

Two-level probit model of  $y$  and covariate  $x$  and random intercepts by `lev2`

```
meprobit y x || lev2:
```

Add random coefficients for  $x$

```
meprobit y x || lev2: x
```

Same as above, but specify that  $y$  records the number of successes from 10 trials

```
meprobit y x || lev2: x, binomial(10)
```

Same as above, but with the number of trials stored in variable `n`

```
meprobit y x || lev2: x, binomial(n)
```

Three-level random-intercept model of  $y$  and covariate  $x$  with `lev2` nested within `lev3`

```
meprobit y x || lev3: || lev2:
```

Two-way crossed random effects by factors `a` and `b`

```
meprobit y x || _all:R.a || b:
```

## Menu

Statistics > Multilevel mixed-effects models > Probit regression

## Syntax

```
meprobit depvar fe_equation [ || re_equation ] [ || re_equation ... ] [ , options ]
```

where the syntax of *fe\_equation* is

```
[ indepvars ] [ if ] [ in ] [ weight ] [ , fe_options ]
```

and the syntax of *re\_equation* is one of the following:

for random coefficients and intercepts

```
levelvar: [ varlist ] [ , re_options ]
```

for random effects among the values of a factor variable in a crossed-effects model

```
levelvar: R.varname
```

*levelvar* is a variable identifying the group structure for the random effects at that level or is `_all` representing one group comprising all observations.

| <i>fe_options</i>                   | Description   |
|-------------------------------------|---|
| Model                               |   |
| <code>noconstant</code>             | suppress constant term from the fixed-effects equation            |
| <code>offset(<i>varname</i>)</code> | include <i>varname</i> in model with coefficient constrained to 1 |
| <code>asis</code>                   | retain perfect predictor variables                                |

| <i>re_options</i>                       | Description   |
|---|---|
| Model                                   |   |
| <code>covariance(<i>vartype</i>)</code> | variance–covariance structure of the random effects     |
| <code>noconstant</code>                 | suppress constant term from the random-effects equation |
| <code>fweight(<i>varname</i>)</code>    | frequency weights at higher levels                      |
| <code>iweight(<i>varname</i>)</code>    | importance weights at higher levels                     |
| <code>pweight(<i>varname</i>)</code>    | sampling weights at higher levels                       |

| <i>options</i>                                      | Description  |
|---|--|
| Model   |  |
| <code><u>binomial</u>(<i>varname</i>   #)</code>    | set binomial trials if data are in binomial form   |
| <code><u>constraints</u>(<i>constraints</i>)</code> | apply specified linear constraints   |
| SE/Robust   |  |
| <code><u>vce</u>(<i>vcetype</i>)</code>             | <i>vcetype</i> may be <code>oim</code> , <code>opg</code> , <code>robust</code> , or <code>cluster</code> <i>clustvar</i>                        |
| Reporting   |  |
| <code><u>level</u>(#)</code>                        | set confidence level; default is <code>level(95)</code>  |
| <code><u>nocnsreport</u></code>                     | do not display constraints   |
| <code><u>notable</u></code>                         | suppress coefficient table   |
| <code><u>noheader</u></code>                        | suppress output header   |
| <code><u>nogroup</u></code>                         | suppress table summarizing groups  |
| <code><u>display_options</u></code>                 | control columns and column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling |
| Integration   |  |
| <code><u>intmethod</u>(<i>intmethod</i>)</code>     | integration method   |
| <code><u>intpoints</u>(#)</code>                    | set the number of integration (quadrature) points for all levels; default is <code>intpoints(7)</code>   |
| Maximization  |  |
| <code><u>maximize_options</u></code>                | control the maximization process; seldom used  |
| <code><u>startvalues</u>(<i>svmethod</i>)</code>    | method for obtaining starting values   |
| <code><u>startgrid</u>[ (<i>gridspec</i>) ]</code>  | perform a grid search to improve starting values   |
| <code><u>noestimate</u></code>                      | do not fit the model; show starting values instead   |
| <code><u>dnnumerical</u></code>                     | use numerical derivative techniques  |
| <code><u>collinear</u></code>                       | keep collinear variables   |
| <code><u>coeflegend</u></code>                      | display legend instead of statistics   |
| <i>vartype</i>                                      | Description  |
| <code><u>independent</u></code>                     | one unique variance parameter per random effect and all covariances 0; the default unless the <code>R.</code> notation is used                   |
| <code><u>exchangeable</u></code>                    | equal variances for random effects and one common pairwise covariance  |
| <code><u>identity</u></code>                        | equal variances for random effects and all covariances 0; the default if the <code>R.</code> notation is used                                    |
| <code><u>unstructured</u></code>                    | all variances and covariances to be distinctly estimated   |
| <code><u>fixed</u>(<i>matname</i>)</code>           | user-selected variances and covariances constrained to specified values; the remaining variances and covariances unrestricted                    |
| <code><u>pattern</u>(<i>matname</i>)</code>         | user-selected variances and covariances constrained to be equal; the remaining variances and covariances unrestricted                            |

| <i>intmethod</i>    | Description   |
|---------------------|---|
| <u>m</u> vaghermite | mean–variance adaptive Gauss–Hermite quadrature; the default unless a crossed random-effects model is fit |
| <u>m</u> caghermite | mode-curvature adaptive Gauss–Hermite quadrature  |
| <u>g</u> hermite    | nonadaptive Gauss–Hermite quadrature  |
| <u>l</u> aplace     | Laplacian approximation; the default for crossed random-effects models                                    |

*indepvars* and *varlist* may contain factor variables; see [U] 11.4.3 Factor variables.

*depvar*, *indepvars*, and *varlist* may contain time-series operators; see [U] 11.4.4 Time-series varlists.

*bayes*, *by*, *collect*, and *svy* are allowed; see [U] 11.1.10 Prefix commands. For more details, see [BAYES] *bayes: meprobit*.

*vce()* and weights are not allowed with the *svy* prefix; see [SVY] *svy*.

*fweights*, *iweights*, and *pweights* are allowed; see [U] 11.1.6 *weight*. Only one type of weight may be specified. Weights are not supported under the Laplacian approximation or for crossed models.

*startvalues()*, *startgrid*, *noestimate*, *dnumerical*, *collinear*, and *coeflegend* do not appear in the dialog box.

See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

## Options

### Model

*noconstant* suppresses the constant (intercept) term and may be specified for the fixed-effects equation and for any of or all the random-effects equations.

*offset(varname)* specifies that *varname* be included in the fixed-effects portion of the model with the coefficient constrained to be 1.

*asis* forces retention of perfect predictor variables and their associated, perfectly predicted observations and may produce instabilities in maximization; see [R] *probit*.

*covariance(vartype)* specifies the structure of the covariance matrix for the random effects and may be specified for each random-effects equation. *vartype* is one of the following: *independent*, *exchangeable*, *identity*, *unstructured*, *fixed(matname)*, or *pattern(matname)*.

*covariance(independent)* covariance structure allows for a distinct variance for each random effect within a random-effects equation and assumes that all covariances are 0. The default is *covariance(independent)* unless a crossed random-effects model is fit, in which case the default is *covariance(identity)*.

*covariance(exchangeable)* structure specifies one common variance for all random effects and one common pairwise covariance.

*covariance(identity)* is short for “multiple of the identity”; that is, all variances are equal and all covariances are 0.

*covariance(unstructured)* allows for all variances and covariances to be distinct. If an equation consists of  $p$  random-effects terms, the unstructured covariance matrix will have  $p(p + 1)/2$  unique parameters.

*covariance(fixed(matname))* and *covariance(pattern(matname))* covariance structures provide a convenient way to impose constraints on variances and covariances of random effects. Each specification requires a *matname* that defines the restrictions placed on variances and covariances. Only elements in the lower triangle of *matname* are used, and row and column names

of *matname* are ignored. A missing value in *matname* means that a given element is unrestricted. In a `fixed(matname)` covariance structure, (co)variance  $(i, j)$  is constrained to equal the value specified in the  $i, j$ th entry of *matname*. In a `pattern(matname)` covariance structure, (co)variances  $(i, j)$  and  $(k, l)$  are constrained to be equal if  $matname[i, j] = matname[k, l]$ .

`fweight(varname)` specifies frequency weights at higher levels in a multilevel model, whereas frequency weights at the first level (the observation level) are specified in the usual manner, for example, `[fw=fwtvar1]`. *varname* can be any valid Stata variable name, and you can specify `fweight()` at levels two and higher of a multilevel model. For example, in the two-level model

```
. mecmd fixed_portion [fw = wt1] || school: ... , fweight(wt2) ...
```

the variable `wt1` would hold the first-level (the observation-level) frequency weights, and `wt2` would hold the second-level (the school-level) frequency weights.

`iweight(varname)` specifies importance weights at higher levels in a multilevel model, whereas importance weights at the first level (the observation level) are specified in the usual manner, for example, `[iw=iwtvar1]`. *varname* can be any valid Stata variable name, and you can specify `iweight()` at levels two and higher of a multilevel model. For example, in the two-level model

```
. mecmd fixed_portion [iw = wt1] || school: ... , iweight(wt2) ...
```

the variable `wt1` would hold the first-level (the observation-level) importance weights, and `wt2` would hold the second-level (the school-level) importance weights.

`pweight(varname)` specifies sampling weights at higher levels in a multilevel model, whereas sampling weights at the first level (the observation level) are specified in the usual manner, for example, `[pw=pwtvar1]`. *varname* can be any valid Stata variable name, and you can specify `pweight()` at levels two and higher of a multilevel model. For example, in the two-level model

```
. mecmd fixed_portion [pw = wt1] || school: ... , pweight(wt2) ...
```

variable `wt1` would hold the first-level (the observation-level) sampling weights, and `wt2` would hold the second-level (the school-level) sampling weights.

`binomial(varname | #)` specifies that the data are in binomial form; that is, *depvar* records the number of successes from a series of binomial trials. This number of trials is given either as *varname*, which allows this number to vary over the observations, or as the constant `#`. If `binomial()` is not specified (the default), *depvar* is treated as Bernoulli, with any nonzero, nonmissing values indicating positive responses.

`constraints(constraints)`; see [\[R\] Estimation options](#).

#### SE/Robust

`vce(vcetype)` specifies the type of standard error reported, which includes types that are derived from asymptotic theory (`oim`, `opg`), that are robust to some kinds of misspecification (`robust`), and that allow for intragroup correlation (`cluster clustvar`); see [\[R\] vce\\_option](#). If `vce(robust)` is specified, robust variances are clustered at the highest level in the multilevel model.

#### Reporting

`level(#)`, `nocnsreport`; see [\[R\] Estimation options](#).

`notable` suppresses the estimation table, either at estimation or upon replay.

`noheader` suppresses the output header, either at estimation or upon replay.

`nogroup` suppresses the display of group summary information (number of groups, average group size, minimum, and maximum) from the output header.

*display\_options*: `nocl`, `nopvalues`, `noomitted`, `vsquish`, `noemptycells`, `baselevels`, `allbaselevels`, `nofvlabel`, `fvwrap(#)`, `fvwrapon(style)`, `cformat(%fmt)`, `pformat(%fmt)`, `sformat(%fmt)`, and `nolstretch`; see [R] [Estimation options](#).

#### Integration

`intmethod(intmethod)` specifies the integration method to be used for the random-effects model. `mvaghermite` performs mean–variance adaptive Gauss–Hermite quadrature; `mcaghermite` performs mode-curvature adaptive Gauss–Hermite quadrature; `ghermite` performs nonadaptive Gauss–Hermite quadrature; and `laplace` performs the Laplacian approximation, equivalent to mode-curvature adaptive Gaussian quadrature with one integration point.

The default integration method is `mvaghermite` unless a crossed random-effects model is fit, in which case the default integration method is `laplace`. The Laplacian approximation has been known to produce biased parameter estimates; however, the bias tends to be more prominent in the estimates of the variance components rather than in the estimates of the fixed effects.

For crossed random-effects models, estimation with more than one quadrature point may be prohibitively intensive even for a small number of levels. For this reason, the integration method defaults to the Laplacian approximation. You may override this behavior by specifying a different integration method.

`intpoints(#)` sets the number of integration points for quadrature. The default is `intpoints(7)`, which means that seven quadrature points are used for each level of random effects. This option is not allowed with `intmethod(laplace)`.

The more integration points, the more accurate the approximation to the log likelihood. However, computation time increases as a function of the number of quadrature points raised to a power equaling the dimension of the random-effects specification. In crossed random-effects models and in models with many levels or many random coefficients, this increase can be substantial.

#### Maximization

*maximize\_options*: `difficult`, `technique(algorithm_spec)`, `iterate(#)`, `[no]log`, `trace`, `gradient`, `showstep`, `hessian`, `showtolerance`, `tolerance(#)`, `ltolerance(#)`, `nrtolerance(#)`, `nonrtolerance`, and `from(init_specs)`; see [R] [Maximize](#). Those that require special mention for `meprobit` are listed below.

`from()` accepts a properly labeled vector of initial values or a list of coefficient names with values. A list of values is not allowed.

The following options are available with `meprobit` but are not shown in the dialog box:

`startvalues(svmethod)`, `startgrid([gridspec])`, `noestimate`, and `dnumerical`; see [ME] [meglm](#).

`collinear`, `coeflegend`; see [R] [Estimation options](#).

## Remarks and examples

Mixed-effects probit regression is probit regression containing both fixed effects and random effects. In longitudinal data and panel data, random effects are useful for modeling intracluster correlation; that is, observations in the same cluster are correlated because they share common cluster-level random effects.

`meprobit` allows for many levels of random effects. However, for simplicity, we here consider the two-level model, where for a series of  $M$  independent clusters, and conditional on a set of fixed effects  $\mathbf{x}_{ij}$  and a set of random effects  $\mathbf{u}_j$ ,

$$\Pr(y_{ij} = 1 | \mathbf{x}_{ij}, \mathbf{u}_j) = H(\mathbf{x}_{ij}\boldsymbol{\beta} + \mathbf{z}_{ij}\mathbf{u}_j) \quad (1)$$

for  $j = 1, \dots, M$  clusters, with cluster  $j$  consisting of  $i = 1, \dots, n_j$  observations. The responses are the binary-valued  $y_{ij}$ , and we follow the standard Stata convention of treating  $y_{ij} = 1$  if `devarij`  $\neq 0$  and treating  $y_{ij} = 0$  otherwise. The  $1 \times p$  row vector  $\mathbf{x}_{ij}$  are the covariates for the fixed effects, analogous to the covariates you would find in a standard probit regression model, with regression coefficients (fixed effects)  $\boldsymbol{\beta}$ . For notational convenience here and throughout this manual entry, we suppress the dependence of  $y_{ij}$  on  $\mathbf{x}_{ij}$ .

The  $1 \times q$  vector  $\mathbf{z}_{ij}$  are the covariates corresponding to the random effects and can be used to represent both random intercepts and random coefficients. For example, in a random-intercept model,  $\mathbf{z}_{ij}$  is simply the scalar 1. The random effects  $\mathbf{u}_j$  are  $M$  realizations from a multivariate normal distribution with mean  $\mathbf{0}$  and  $q \times q$  variance matrix  $\boldsymbol{\Sigma}$ . The random effects are not directly estimated as model parameters but are instead summarized according to the unique elements of  $\boldsymbol{\Sigma}$ , known as variance components. One special case of (1) places  $\mathbf{z}_{ij} = \mathbf{x}_{ij}$ , so that all covariate effects are essentially random and distributed as multivariate normal with mean  $\boldsymbol{\beta}$  and variance  $\boldsymbol{\Sigma}$ .

Finally, because this is probit regression,  $H(\cdot)$  is the standard normal cumulative distribution function, which maps the linear predictor to the probability of a success ( $y_{ij} = 1$ ) with  $H(v) = \Phi(v)$ .

Model (1) may also be stated in terms of a latent linear response, where only  $y_{ij} = I(y_{ij}^* > 0)$  is observed for the latent

$$y_{ij}^* = \mathbf{x}_{ij}\boldsymbol{\beta} + \mathbf{z}_{ij}\mathbf{u}_j + \epsilon_{ij}$$

The errors  $\epsilon_{ij}$  are distributed as a standard normal with mean 0 and variance 1 and are independent of  $\mathbf{u}_j$ .

Below we present two short examples of mixed-effects probit regression; refer to [ME] `me` and [ME] `meglm` for examples of other random-effects models. A two-level probit model can also be fit using `xtprobit` with the `re` option; see [XT] `xtprobit`. In the absence of random effects, mixed-effects probit regression reduces to standard probit regression; see [R] `probit`.

### ► Example 1: Two-level random-intercept model

Ng et al. (2006) analyzed a subsample of data from the 1989 Bangladesh fertility survey (Huq and Cleland 1990), which polled 1,934 Bangladeshi women on their use of contraception. The women sampled were from 60 districts, identified by the variable `district`. Each district contained either urban or rural areas (variable `urban`) or both. The variable `c_use` is the binary response, with a value of 1 indicating contraceptive use. Other covariates include mean-centered `age` and three indicator variables recording number of children.

```

. use https://www.stata-press.com/data/r18/bangladesh
(Bangladesh Fertility Survey, 1989)
. meprobit c_use i.urban age i.children || district:
Fitting fixed-effects model:
Iteration 0: Log likelihood = -1228.8313
Iteration 1: Log likelihood = -1228.2466
Iteration 2: Log likelihood = -1228.2466
Refining starting values:
Grid node 0: Log likelihood = -1237.3973
Fitting full model:
Iteration 0: Log likelihood = -1237.3973 (not concave)
Iteration 1: Log likelihood = -1221.2111 (not concave)
Iteration 2: Log likelihood = -1207.4451
Iteration 3: Log likelihood = -1206.7002
Iteration 4: Log likelihood = -1206.5346
Iteration 5: Log likelihood = -1206.5336
Iteration 6: Log likelihood = -1206.5336
Mixed-effects probit regression
Group variable: district
Number of obs      =      1,934
Number of groups   =         60
Obs per group:
    min =          2
    avg =         32.2
    max =         118
Integration method: mvaghermite
Integration pts.   =          7
Wald chi2(5)      =        115.36
Prob > chi2       =         0.0000
Log likelihood = -1206.5336

```

| c_use       | Coefficient | Std. err. | z      | P> z  | [95% conf. interval] |           |
|-------------|-------------|-----------|--------|-------|----------------------|-----------|
| urban       |             |           |        |       |                      |           |
| Urban       | .4490191    | .0727176  | 6.17   | 0.000 | .3064953             | .5915429  |
| age         | -.0162203   | .0048005  | -3.38  | 0.001 | -.0256291            | -.0068114 |
| children    |             |           |        |       |                      |           |
| 1 child     | .674377     | .0947829  | 7.11   | 0.000 | .488606              | .8601481  |
| 2 children  | .8281581    | .1048136  | 7.90   | 0.000 | .6227272             | 1.033589  |
| 3 or more.. | .8137876    | .1073951  | 7.58   | 0.000 | .6032972             | 1.024278  |
| _cons       | -1.02799    | .0870307  | -11.81 | 0.000 | -1.198567            | -.8574132 |
| district    |             |           |        |       |                      |           |
| var(_cons)  | .0798719    | .026886   |        |       | .0412921             | .1544972  |

```
LR test vs. probit model: chibar2(01) = 43.43      Prob >= chibar2 = 0.0000
```

Probit regression coefficients are most commonly interpreted in terms of partial effects, as we demonstrate in [example 1](#) of [\[ME\] meprobit postestimation](#). For now, we only note that urban women and women with more children are more likely to use contraceptives and that contraceptive use decreases with age. The estimated variance of the random intercept at the district level,  $\hat{\sigma}^2$ , is 0.08 with standard error 0.03. The reported likelihood-ratio test shows that there is enough variability between districts to favor a mixed-effects probit regression over an ordinary probit regression; see [Distribution theory for likelihood-ratio test](#) in [\[ME\] me](#) for a discussion of likelihood-ratio testing of variance components.

## ▷ Example 2: Three-level random-intercept model

Rabe-Hesketh, Touloupoulou, and Murray (2001) analyzed data from a study that measured the cognitive ability of patients with schizophrenia compared with their relatives and control subjects. Cognitive ability was measured as the successful completion of the “Tower of London”, a computerized task, measured at three levels of difficulty. For all but one of the 226 subjects, there were three measurements (one for each difficulty level). Because patients’ relatives were also tested, a family identifier, `family`, was also recorded.

We fit a probit model with response `dt1m`, the indicator of cognitive function, and with covariates `difficulty` and a set of indicator variables for `group`, with the controls (`group==1`) being the base category. We also allow for random effects due to families and due to subjects within families. The first is a random intercept (constant only) at the `family` level, and the second is a random intercept at the `subject` level. The order in which these are specified (from left to right) is significant—`meprobit` assumes that `subject` is nested within `family`. The equations are separated by `||`.

```
. use https://www.stata-press.com/data/r18/towerlondon
(Tower of London data)

. meprobit dtlm difficulty i.group || family: || subject:

Fitting fixed-effects model:

Iteration 0:  Log likelihood = -317.11238
Iteration 1:  Log likelihood = -314.50535
Iteration 2:  Log likelihood = -314.50121
Iteration 3:  Log likelihood = -314.50121

Refining starting values:

Grid node 0:  Log likelihood = -326.18533

Fitting full model:

Iteration 0:  Log likelihood = -326.18533 (not concave)
Iteration 1:  Log likelihood = -313.16256 (not concave)
Iteration 2:  Log likelihood = -308.47528
Iteration 3:  Log likelihood = -305.02228
Iteration 4:  Log likelihood = -304.88972
Iteration 5:  Log likelihood = -304.88845
Iteration 6:  Log likelihood = -304.88845

Mixed-effects probit regression                Number of obs    =          677
```

Grouping information

| Group variable | No. of groups | Observations per group |         |         |
|----------------|---------------|------------------------|---------|---------|
|                |               | Minimum                | Average | Maximum |
| family         | 118           | 2                      | 5.7     | 27      |
| subject        | 226           | 2                      | 3.0     | 3       |

```
Integration method: mvaghermite                Integration pts. =           7
Wald chi2(3) = 83.28
Log likelihood = -304.88845                    Prob > chi2 = 0.0000
```

| dtlm       | Coefficient | Std. err. | z     | P> z  | [95% conf. interval] |           |
|------------|-------------|-----------|-------|-------|----------------------|-----------|
| difficulty | -.9329891   | .1037376  | -8.99 | 0.000 | -1.136311            | -.7296672 |
| group      |             |           |       |       |                      |           |
| 2          | -.1632243   | .204265   | -0.80 | 0.424 | -.5635763            | .2371276  |
| 3          | -.6220196   | .228063   | -2.73 | 0.006 | -1.069015            | -.1750244 |
| _cons      | -.8405154   | .1597223  | -5.26 | 0.000 | -1.153565            | -.5274654 |
| family     |             |           |       |       |                      |           |
| var(_cons) | .2120948    | .1736281  |       |       | .0426292             | 1.055244  |
| family>    |             |           |       |       |                      |           |
| subject    |             |           |       |       |                      |           |
| var(_cons) | .3559141    | .219331   |       |       | .106364              | 1.190956  |

```
LR test vs. probit model: chi2(2) = 19.23                Prob > chi2 = 0.0001
```

Note: LR test is conservative and provided only for reference.

We see that we have 226 subjects from 118 families. After adjusting for the random-effects structure, the probability of successful completion of the Tower of London decreases dramatically as the level of difficulty increases. Also, people with schizophrenia (group==3) tended not to perform as well as the control subjects.

The above extends to models with more than two levels of nesting by adding more random-effects equations, each separated by `||`.

## Stored results

`meprobit` stores the following in `e()`:

### Scalars

|                            |   |
|----------------------------|---|
| <code>e(N)</code>          | number of observations                    |
| <code>e(k)</code>          | number of parameters                      |
| <code>e(k_dv)</code>       | number of dependent variables             |
| <code>e(k_eq)</code>       | number of equations in <code>e(b)</code>  |
| <code>e(k_eq_model)</code> | number of equations in overall model test |
| <code>e(k_f)</code>        | number of fixed-effects parameters        |
| <code>e(k_r)</code>        | number of random-effects parameters       |
| <code>e(k_rs)</code>       | number of variances                       |
| <code>e(k_rc)</code>       | number of covariances                     |
| <code>e(df_m)</code>       | model degrees of freedom                  |
| <code>e(ll)</code>         | log likelihood                            |
| <code>e(N_clust)</code>    | number of clusters                        |
| <code>e(chi2)</code>       | $\chi^2$                                  |
| <code>e(p)</code>          | $p$ -value for model test                 |
| <code>e(ll_c)</code>       | log likelihood, comparison model          |
| <code>e(chi2_c)</code>     | $\chi^2$ , comparison test                |
| <code>e(df_c)</code>       | degrees of freedom, comparison test       |
| <code>e(p_c)</code>        | $p$ -value for comparison test            |
| <code>e(rank)</code>       | rank of <code>e(V)</code>                 |
| <code>e(ic)</code>         | number of iterations                      |
| <code>e(rc)</code>         | return code                               |
| <code>e(converged)</code>  | 1 if converged, 0 otherwise               |

### Macros

|                                       |   |
|---------------------------------------|---|
| <code>e(cmd)</code>                   | <code>megl</code>   |
| <code>e(cmd2)</code>                  | <code>meprobit</code>   |
| <code>e(cmdline)</code>               | command as typed  |
| <code>e(depvar)</code>                | name of dependent variable  |
| <code>e(wtype)</code>                 | weight type   |
| <code>e(wexp)</code>                  | weight expression (first-level weights)   |
| <code>e(fweight<math>k</math>)</code> | <code>fweight</code> variable for $k$ th highest level, if specified                                |
| <code>e(iweight<math>k</math>)</code> | <code>iweight</code> variable for $k$ th highest level, if specified                                |
| <code>e(pweight<math>k</math>)</code> | <code>pweight</code> variable for $k$ th highest level, if specified                                |
| <code>e(covariates)</code>            | list of covariates  |
| <code>e(ivars)</code>                 | grouping variables  |
| <code>e(model)</code>                 | <code>probit</code>   |
| <code>e(title)</code>                 | title in estimation output  |
| <code>e(link)</code>                  | <code>probit</code>   |
| <code>e(family)</code>                | <code>bernoulli</code> or <code>binomial</code>   |
| <code>e(clustvar)</code>              | name of cluster variable  |
| <code>e(offset)</code>                | offset  |
| <code>e(binomial)</code>              | binomial number of trials   |
| <code>e(intmethod)</code>             | integration method  |
| <code>e(n_quad)</code>                | number of integration points  |
| <code>e(chi2type)</code>              | Wald; type of model $\chi^2$  |
| <code>e(vce)</code>                   | <code>vcetype</code> specified in <code>vce()</code>  |
| <code>e(vcetype)</code>               | title used to label Std. err.   |
| <code>e(opt)</code>                   | type of optimization  |
| <code>e(which)</code>                 | <code>max</code> or <code>min</code> ; whether optimizer is to perform maximization or minimization |
| <code>e(ml_method)</code>             | type of <code>ml</code> method  |
| <code>e(user)</code>                  | name of likelihood-evaluator program  |
| <code>e(technique)</code>             | maximization technique  |
| <code>e(datasignature)</code>         | the checksum  |
| <code>e(datasignaturevars)</code>     | variables used in calculation of checksum   |

|                              |  |
|------------------------------|--|
| <code>e(properties)</code>   | <code>b V</code>   |
| <code>e(estat_cmd)</code>    | program used to implement <code>estat</code>                   |
| <code>e(predict)</code>      | program used to implement <code>predict</code>                 |
| <code>e(marginsnotok)</code> | predictions disallowed by <code>margins</code>                 |
| <code>e(marginswtype)</code> | weight type for <code>margins</code>                           |
| <code>e(marginswexp)</code>  | weight expression for <code>margins</code>                     |
| <code>e(asbalanced)</code>   | factor variables <code>fvset</code> as <code>asbalanced</code> |
| <code>e(asobserved)</code>   | factor variables <code>fvset</code> as <code>asobserved</code> |
| Matrices                     |  |
| <code>e(b)</code>            | coefficient vector   |
| <code>e(Cns)</code>          | constraints matrix   |
| <code>e(ilog)</code>         | iteration log (up to 20 iterations)                            |
| <code>e(gradient)</code>     | gradient vector  |
| <code>e(N_g)</code>          | group counts   |
| <code>e(g_min)</code>        | group-size minimums  |
| <code>e(g_avg)</code>        | group-size averages  |
| <code>e(g_max)</code>        | group-size maximums  |
| <code>e(V)</code>            | variance-covariance matrix of the estimators                   |
| <code>e(V_modelbased)</code> | model-based variance   |
| Functions                    |  |
| <code>e(sample)</code>       | marks estimation sample  |

In addition to the above, the following is stored in `r()`:

|                       |  |
|-----------------------|--|
| Matrices              |  |
| <code>r(table)</code> | matrix containing the coefficients with their standard errors, test statistics, <i>p</i> -values, and confidence intervals |

Note that results stored in `r()` are updated when the command is replayed and will be replaced when any *r*-class command is run after the estimation command.

## Methods and formulas

`meprobit` is a convenience command for `meglm` with a `probit` link and a `bernoulli` or `binomial` family; see [ME] `meglm`.

Model (1) assumes Bernoulli data, a special case of the binomial. Because binomial data are also supported by `meprobit` (option `binomial()`), the methods presented below are in terms of the more general binomial mixed-effects model.

For a two-level binomial model, consider the response  $y_{ij}$  as the number of successes from a series of  $r_{ij}$  Bernoulli trials (replications). For cluster  $j$ ,  $j = 1, \dots, M$ , the conditional distribution of  $\mathbf{y}_j = (y_{j1}, \dots, y_{jn_j})'$ , given a set of cluster-level random effects  $\mathbf{u}_j$ , is

$$\begin{aligned} f(\mathbf{y}_j | \mathbf{u}_j) &= \prod_{i=1}^{n_j} \left[ \binom{r_{ij}}{y_{ij}} \{\Phi(\boldsymbol{\eta}_{ij})\}^{y_{ij}} \{1 - \Phi(\boldsymbol{\eta}_{ij})\}^{r_{ij} - y_{ij}} \right] \\ &= \exp \left( \sum_{i=1}^{n_j} \left[ y_{ij} \log \{\Phi(\boldsymbol{\eta}_{ij})\} - (r_{ij} - y_{ij}) \log \{\Phi(-\boldsymbol{\eta}_{ij})\} + \log \binom{r_{ij}}{y_{ij}} \right] \right) \end{aligned}$$

for  $\boldsymbol{\eta}_{ij} = \mathbf{x}_{ij}\boldsymbol{\beta} + \mathbf{z}_{ij}\mathbf{u}_j + \text{offset}_{ij}$ .

Defining  $\mathbf{r}_j = (r_{j1}, \dots, r_{jn_j})'$  and

$$c(\mathbf{y}_j, \mathbf{r}_j) = \sum_{i=1}^{n_j} \log \binom{r_{ij}}{y_{ij}}$$

where  $c(\mathbf{y}_j, \mathbf{r}_j)$  does not depend on the model parameters, we can express the above compactly in matrix notation,

$$f(\mathbf{y}_j | \mathbf{u}_j) = \exp [\mathbf{y}'_j \log \{\Phi(\boldsymbol{\eta}_j)\} - (\mathbf{r}_j - \mathbf{y}_j)' \log \{\Phi(-\boldsymbol{\eta}_j)\} + c(\mathbf{y}_j, \mathbf{r}_j)]$$

where  $\boldsymbol{\eta}_j$  is formed by stacking the row vectors  $\boldsymbol{\eta}_{ij}$ . We extend the definitions of  $\Phi(\cdot)$ ,  $\log(\cdot)$ , and  $\exp(\cdot)$  to be vector functions where necessary.

Because the prior distribution of  $\mathbf{u}_j$  is multivariate normal with mean  $\mathbf{0}$  and  $q \times q$  variance matrix  $\boldsymbol{\Sigma}$ , the likelihood contribution for the  $j$ th cluster is obtained by integrating  $\mathbf{u}_j$  out of the joint density  $f(\mathbf{y}_j, \mathbf{u}_j)$ ,

$$\begin{aligned} \mathcal{L}_j(\boldsymbol{\beta}, \boldsymbol{\Sigma}) &= (2\pi)^{-q/2} |\boldsymbol{\Sigma}|^{-1/2} \int f(\mathbf{y}_j | \mathbf{u}_j) \exp(-\mathbf{u}'_j \boldsymbol{\Sigma}^{-1} \mathbf{u}_j / 2) d\mathbf{u}_j \\ &= \exp\{c(\mathbf{y}_j, \mathbf{r}_j)\} (2\pi)^{-q/2} |\boldsymbol{\Sigma}|^{-1/2} \int \exp\{h(\boldsymbol{\beta}, \boldsymbol{\Sigma}, \mathbf{u}_j)\} d\mathbf{u}_j \end{aligned} \quad (2)$$

where

$$h(\boldsymbol{\beta}, \boldsymbol{\Sigma}, \mathbf{u}_j) = \mathbf{y}'_j \log \{\Phi(\boldsymbol{\eta}_j)\} - (\mathbf{r}_j - \mathbf{y}_j)' \log \{\Phi(-\boldsymbol{\eta}_j)\} - \mathbf{u}'_j \boldsymbol{\Sigma}^{-1} \mathbf{u}_j / 2$$

and for convenience, in the arguments of  $h(\cdot)$  we suppress the dependence on the observable data  $(\mathbf{y}_j, \mathbf{r}_j, \mathbf{X}_j, \mathbf{Z}_j)$ .

The integration in (2) has no closed form and thus must be approximated; see *Methods and formulas* in [ME] **meglm** for details.

**meprobit** supports multilevel weights and survey data; see *Methods and formulas* in [ME] **meglm** for details.

## References

- Huq, N. M., and J. Cleland. 1990. *Bangladesh Fertility Survey 1989 (Main Report)*. National Institute of Population Research and Training.
- Ng, E. S.-W., J. R. Carpenter, H. Goldstein, and J. Rasbash. 2006. Estimation in generalised linear mixed models with binary outcomes by simulated maximum likelihood. *Statistical Modelling* 6: 23–42. <https://doi.org/10.1191/1471082X06st106oa>.
- Rabe-Hesketh, S., T. Touloupoulou, and R. M. Murray. 2001. Multilevel modeling of cognitive function in schizophrenic patients and their first degree relatives. *Multivariate Behavioral Research* 36: 279–298. [https://doi.org/10.1207/S15327906MBR3602\\_07](https://doi.org/10.1207/S15327906MBR3602_07).

## Also see

- [ME] **meprobit postestimation** — Postestimation tools for meprobit
- [ME] **mecloglog** — Multilevel mixed-effects complementary log–log regression
- [ME] **melogit** — Multilevel mixed-effects logistic regression
- [ME] **me** — Introduction to multilevel mixed-effects models
- [BAYES] **bayes: meprobit** — Bayesian multilevel probit regression
- [SEM] **Intro 5** — Tour of models (*Multilevel mixed-effects models*)
- [SVY] **svy estimation** — Estimation commands for survey data

[XT] **xtprobit** — Random-effects and population-averaged probit models

[U] **20 Estimation and postestimation commands**

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.

